

StrategyBuilder™

Important Information and Disclaimer:

All support, education and training services and materials on the TradeStation website are for informational purposes and to help customers learn more about how to use the power of TradeStation software and services. No type of trading or investment advice is being made, given or in any manner provided by any TradeStation affiliate.

This material may also discuss in detail how TradeStation is designed to help you develop, test and implement trading strategies. However, TradeStation does not provide or suggest trading strategies. We offer you unique tools to help you design your own strategies and look at how they could have performed in the past. While we believe this is very valuable information, we caution you that simulated past performance of a trading strategy is no guarantee of its future performance or success. We also do not recommend or solicit the purchase or sale of any particular securities or derivative products. Any symbols referenced are used only for the purposes of the demonstration, as an example—not a recommendation.

Finally, this material may discuss automated electronic order placement and execution. Please note that even though TradeStation has been designed to automate your trading strategies and deliver timely order placement, routing and execution, these things, as well as access to the system itself, may at times be delayed or even fail due to market volatility, quote delays, system and software errors, Internet traffic, outages and other factors.

TradeStation Group, Inc. Affiliates: All proprietary technology in TradeStation is owned by TradeStation Technologies, Inc. Equities, equities options, and commodity futures products and services are offered by TradeStation Securities, Inc. (Member NYSE, FINRA, NFA and SIPC). TradeStation Securities, Inc.'s SIPC coverage is available only for equities and equities options accounts. Forex products and services are offered by TradeStation Forex, a division of IBFX, Inc. (Member NFA).

Copyright © 2001-2013 TradeStation Group, Inc.

****StrategyBuilder™ (Charting Strategy)**

Features

- Analysis Focus: Trend Following
 - Markets: All
 - Time Horizon: Mid-Term to Long-Term
-

Summary

The **StrategyBuilder™ Strategy for TradeStation Charting is a tool that uses optimization to help build a long-only trending strategy based on a pre-determined set of strategy entry and exit rules. By adjusting or optimizing the strategy inputs, you can easily build a multi-conditional strategy tailored to the symbol and interval in the chart, without any programming.

The structure of this strategy allows you to select one of four entry-filter conditions, one of four entry-trigger conditions, and one of four exit conditions. Both filter and trigger conditions must be true on the same bar in order to generate a long buy order.

There are four exit criteria in the strategy, but any of the built-in exit component strategies that come with charting can also be inserted into the chart and will work in conjunction with the StrategyBuilder™ exits.

In addition, not only can you select the strategy order rules, but you can also select and optimize the various indicator input lengths and criteria. This is where TradeStation's genetic optimization can really help you manage and speed up large optimization runs.

The entry and exit rules are based on standard technical analysis rules used for long-term trending strategies. The strategy entry and exit rules are describe below in the input section.


This strategy is designed as a prototype that can be used as a template for other strategy-building tools.

Important:

Keep in mind that StrategyBuilder™ is designed to show you what set of strategy rules and inputs would have performed the best during the historical simulation, and any future performance may be very different. Past performance does not guarantee future success. The very nature of optimization is to curve-fit the results to the historical data set.

You should always forward-test your strategies and trading ideas in a simulated environment before risking your money.

**StrategyBuilder™ (Charting Strategy)



Simulated past performance does not guarantee future success.

The **StrategyBuilder™ (Charting Strategy) displays the buy and sell signals based on the strategy inputs rule codes. The strategy rule codes and input values are displayed in an EasyLanguage form docked at the bottom of the chart. This form displays the current input values set in the strategy.

The sample workspace (StrategyBuilder™.tsw) has a single chart with the **StrategyBuilder™ strategy and StrategyEquity indicator applied.

Strategy Inputs:

The first set of three inputs are the strategy entry and exit rules; each strategy rule will have additional inputs that adjust the technical calculations; and all inputs can be optimized.

FilterCriteria (0) - This is the first entry condition of the strategy.

- 0 = No Filter, 1 = FastMovAvg > SlowMovAvg,
- 2 = PercentR > inputValue, 3 = DMI+ > DMI-

TriggerCriteria (0) - This is the second entry condition of the strategy.

- 0 = On Close, 1 = FastMovAvg Crosses Above SlowMovAvg,
- 2 = Low of the Bar Crosses Above MovAvg of the Highs, 3 = Momentum Crosses Above 0

ExitCriteria (0) - This is the exit condition of the strategy.

- 0 = BarsSinceEntry, 1 = AvgTrueRange Trailing Stop
- 2 = High of the Bar Crosses Below MovAvg of the Lows, 3 = Momentum Crosses Below 0

The next set of inputs control the actual indicator rule calculations.

Filter Inputs:

- FC1_MaLen1(50) - Fast MovAvg Length for FilterCriteria(1)
- FC1_MaLen2(200) - Slow MovAvg Length for FilterCriteria(1)
- FC2_PctRLen(50) - PercentR Length for FilterCriteria(2)
- FC2_PctRTreshold(50) - PercentR Threshold Value for FilterCriteria(2)
- FC3_DMILen(50) - DMI + and - Length for FilterCriteria(3)

Trigger Inputs:


- TC1_MaLen1(20) - Fast MovAvg Length for TriggerCriteria(1)
- TC1_MaLen2(50) - Slow MovAvg Length for TriggerCriteria(1)
- TC2_MaLen(20) - MovAvg Length for TriggerCriteria(2)
- TC3_MOLen(20) - Momentum Length for TriggerCriteria(3)

Exit Inputs:

- XC0_BSE(20) - Bars to Exit from Entry Bar for ExitCriteria(0)
- XC1_ATRLen(20) - AvgTrueRange Length for ExitCriteria(1)
- XC1_ATRs(2) - Number of ATRs for trail for ExitCriteria(1)
- XC2_MALen(20) - MovAvg Length for ExitCriteria(2)
- XC3_MOLen(20) - Momentum Length for ExitCriteria(3)


Optimization Techniques

With so many inputs to optimize, you may want to first just optimize the strategy rules, to get a feel for the process. Go to Format Strategy - Select Format - and then go to the Inputs tab, select FilterCriteria and press the optimize button. Set the Start, Stop, and Increment values to 0, 3, and 1. This will optimize on the four possible filter states. Now do this for the Trigger Criteria and Exit Criteria. Click OK, then Optimize. This will result in 64 different strategy combinations.



When completed you can go to the Optimization Report to see all the results, and also view the Performance Report for the final selected set of inputs. Both reports are available from the View menu in Charting.


Next, you may want to optimize one or more of the indicator calculation inputs. Keep in mind that the more inputs you optimize at the same time, and the number of tests you specify, the longer the optimization will take, (sometimes many hours or longer). The trick is to be smart about the optimized settings and not set the optimization increment too fine.



For example, by setting the increment here to 10 instead of 1, you can reduce the number of optimization runs significantly. This savings is compounded if you are optimizing several inputs at the same time.

Genetic Optimization

Before running any large optimization, you can choose to use a Genetic method to reduce the time it takes for large numbers of scenarios. Be sure to Click Advanced and then the Suggest button to set the threshold settings. Remember that Genetic optimizations do not calculate all possible strategy parameter combinations but will use an algorithm based on a kind of natural selection to narrow the combinations.



You can learn more about Genetic optimization from the TradeStation online Help.

Learn More about Strategy Trading

TradeStation University's School of Strategy Trading offers educational tools to learn about strategy trading, development, back-testing, performance evaluation, optimization, and strategy automation. There are a variety of tools and media at all levels that allow you to tap into the power of strategy testing at your own pace.

<http://www.tradestation.com/en/education/university/school-of-strategy-trading>

The EasyLanguage code is provided below for those of you who would like to extend or adjust the strategy rules.

****StrategyBuilder™ (EL CODE)**

```
using elsystem.windows.forms;
using elsystem.drawing;

inputs:
    FilterCriteria(0),
    TriggerCriteria(0),
    ExitCriteria(0),

    FC1_MaLen1(50), FC1_MaLen2(200), FC2_PctRLen(50), FC2_PctRTreshold(50),
    FC3_DMILen(50),

    TC1_MaLen1(20), TC1_MaLen2(50),      TC2_MaLen(20),      TC3_MOLen(20),

    XC0_BSE(20), XC1_ATRLen(20), XC1_ATRs(2), XC2_MALen(20), XC3_MOLen(20);

vars:
    FC1_MaVal1(0), FC1_MaVal2(0), FC2_PctrVal(0), FC3_DMIPlus(0),
    FC3_DMIMinus(0),
    TC1_MaVal1(0), TC1_MaVal2(0), TC2_MaVal(0), TC3_MOVal(0),
    XC1_ATR(0), XC1_ATRVal(0), XC2_MAVal(0), XC3_MOVal(0),
    FCCond(False), MP(0);

vars: Form SBForm( Null ), Label SBLLabel1(Null), Label SBLLabel2(Null ),
    Label SBLLabel3(Null), Label SBLLabel4(Null), Label SBLLabel5(Null),
    Font SBTitle(Null);

method void AnalysisTechnique_Initialized( elsystem.Object sender,
    elsystem.InitializedEventArgs args )
var: string tempstr;
begin

If Getappinfo(aiOptimizing) = 0 then begin
    SBForm = form.create("StrategyBuilder™", 1080, 105);
    SBLLabel1 = Label.Create("Strategy Builder (Long Entry Only)", 720, 18);
    SBLLabel1.Location( 1,1 );
    SBTitle = Font.Create("",8,fontstyle.bold);
    SBLLabel1.Font = SBTitle;

    SBLLabel2 = Label.Create("", 720, 18);
    tempstr = "Current Input Settings:( Filter Criteria = " +
    numtostr(FilterCriteria,0) +
    " / Trigger Criteria = " + numtostr(TriggerCriteria,0) +
    " / Exit Criteria = " + numtostr(ExitCriteria,0) + " )";
    SBLLabel2.Text = tempstr;
    SBLLabel2.Location( 5,19 );
    SBLLabel2.Font = SBTitle;
```



```

tempstr = "Filter Inputs: 0 = No Filter, " +
"1 = " + numtostr(FC1_MaLen1,0) + " MovAvg > " +
          numtostr(FC1_MaLen2,0) + " MovAvg, " +
"2 = " + numtostr(FC2_PctRLen,0) + " %R > " +
          numtostr(FC2_PctRTreshold,0) + ", " +
"3 = " + numtostr(FC3_DMILen,0) + " DMI+ > " +
          numtostr(FC3_DMILen,0) + " DMI-";

SLabel3 = Label.Create("", 720, 18);
SLabel3.Text = tempstr;
SLabel3.Location( 10,38 );

tempstr = "Trigger Inputs: 0 = Bar Close, " +
"1 = " + numtostr(TC1_MaLen1,0) + " MovAvg Xabove " +
          numtostr(TC1_MaLen2,0) + " MovAvg, " +
"2 = Low Xabove " + numtostr(TC2_MaLen,0) + " MovAvg ofHighs, " +
"3 = " + numtostr(TC3_MOLen,0) + " MomentumAvg Xabove 0";

SLabel4 = Label.Create("", 720, 18);
SLabel4.Text = tempstr;
SLabel4.Location( 10,57 );

tempstr = "Exit Inputs: 0 = " +
          numtostr(XC0_BSE,0) + " BarsSinceEntry " +
"1 = " + numtostr(XC1_ATRLen,0) + " ATR Trails " +
          numtostr(XC1_ATRs,0) + " ATRs, " +
"2 = High Xbelow " + numtostr(XC2_MALen,0) + " MovAvg ofLows, " +
"3 = " + numtostr(XC3_MOLen,0) + " MomentumAvg Xbelow 0";

SLabel5 = Label.Create("", 720, 18);
SLabel5.Text = tempstr;
SLabel5.Location( 10,76 );

SBForm.AddControl(SLabel1);
SBForm.AddControl(SLabel2);
SBForm.AddControl(SLabel3);
SBForm.AddControl(SLabel4);
SBForm.AddControl(SLabel5);
SBForm.Dock = DockStyle.bottom;
SBForm.show();
end;

end;

If FilterCriteria = 1 then begin
    FC1_MaVal1 = Average(Close, FC1_MaLen1);
    FC1_MaVal2 = Average(Close, FC1_MaLen2);
end;
If FilterCriteria = 2 then
    FC2_PctRVal = PercentR(FC2_PctRLen);
If FilterCriteria = 3 then begin
    FC3_DMIPlus = DMIPlus(FC3_DMILen);
    FC3_DMIMinus = DMIMinus(FC3_DMILen);
end;
If TriggerCriteria = 1 then begin
    TC1_MaVal1 = Average(Close, TC1_MaLen1);
    TC1_MaVal2 = Average(Close, TC1_MaLen2);
end;

```

```

If TriggerCriteria = 2 then
    TC2_MaVal = Average(Close, TC2_MaLen);
If TriggerCriteria = 3 then
    TC3_MOVal = Average(Momentum(Close, TC3_MOLen),3);
If ExitCriteria = 1 then begin
    XC1_ATR = AvgTrueRange(XC1_ATRLen);
    XC1_ATRVal = XC1_ATR * XC1_ATRs;
end;
If ExitCriteria = 2 then
    XC2_MaVal = Average(Close, XC2_MALen);
If ExitCriteria = 3 then
    XC3_MOVal = Average(Momentum(Close, XC3_MOLen),3);

SetStopShare;
MP = MarketPosition;

Switch(FilterCriteria)
Begin
    Case 0: FCCond = TRUE;
    Case 1: FCCond = FC1_MaVal1 > FC1_MaVal2;
    Case 2: FCCond = FC2_PctRVal > FC2_PctRTreshold;
    Case 3: FCCond = FC3_DMIPplus > FC3_DMIMinus;
    Default: FCCond = FALSE;
end;
Switch(TriggerCriteria)
Begin
    Case 0: if MP = 0 AND FCCond then Buy("LE") this bar on Close;
    Case 1: if MP = 0 AND FCCond AND TC1_MaVal1 > TC1_MaVal2 then
        Buy("Avgx LE") this bar on Close;
    Case 2: if MP = 0 AND FCCond AND Low > TC2_MaVal then
        Buy("Lx LE") this bar on Close;
    Case 3: if MP = 0 AND FCCond AND TC3_MOVal > 0 then
        Buy("Mox LE") this bar on Close;
    Default: FCCond = FALSE;
end;
Switch(ExitCriteria)
Begin
    Case 0: If MP = 1 AND BarsSinceEntry >= XC0_BSE then
        Sell("BSE LX") this bar on Close;
    Case 1: If MP = 1 then Setdollartrailing(XC1_ATRVal * BigPointValue);
    Case 2: If MP = 1 AND High Crosses Below XC2_MaVal then
        Sell("Hx LX") this bar on Close;
    Case 3: If MP = 1 AND XC3_MOVal Crosses Below 0 then
        Sell("Mox LX") this bar on Close;
    Default: SetExitonClose;
end;

```

Important Information and Disclaimer:

All support, education and training services and materials on the TradeStation website are for informational purposes and to help customers learn more about how to use the power of TradeStation software and services. No type of trading or investment advice is being made, given or in any manner provided by any TradeStation affiliate.

This material may also discuss in detail how TradeStation is designed to help you develop, test and implement trading strategies. However, TradeStation does not provide or suggest trading strategies. We offer you unique tools to help you design your own strategies and look at how they could have performed in the past. While we believe this is very valuable information, we caution you that simulated past performance of a trading strategy is no guarantee of its future performance or success. We also do not recommend or solicit the purchase or sale of any particular securities or derivative products. Any symbols referenced are used only for the purposes of the demonstration, as an example—not a recommendation.

Finally, this material may discuss automated electronic order placement and execution. Please note that even though TradeStation has been designed to automate your trading strategies and deliver timely order placement, routing and execution, these things, as well as access to the system itself, may at times be delayed or even fail due to market volatility, quote delays, system and software errors, Internet traffic, outages and other factors.

TradeStation Group, Inc. Affiliates: All proprietary technology in TradeStation is owned by TradeStation Technologies, Inc. Equities, equities options, and commodity futures products and services are offered by TradeStation Securities, Inc. (Member NYSE, FINRA, NFA and SIPC). TradeStation Securities, Inc.'s SIPC coverage is available only for equities and equities options accounts. Forex products and services are offered by TradeStation Forex, a division of IBFX, Inc. (Member NFA).

Copyright © 2001-2013 TradeStation Group, Inc.
